# Chapter 4

# Cell Cycle

## 4.1 Modeling conventions

This short summary is based on Tyson and Novak (2001) (See (4) and a readable survey for modelers in (5).)

Let $C$ denote the concentration of a protein participating in one of the reactions, and suppose that $Q$ is the concentration of a regulatory substance that binds to $C$. Then the standard assumption is that

$$\frac{\partial C}{\partial t} = k_{\text{syn}}(\text{substrate}) - k_{\text{decay}}C - k_{\text{assoc}}CQ. \tag{4.1}$$

Here $k_{\text{syn}}$ represents a rate of protein synthesis from amino acids, $k_{\text{assoc}}$ is rate of association of $C$ with $Q$, and $k_{\text{degrd}}$ is a rate of degradation of $C$ by $Q$.

In the case where a regulatory enzymes convert an inactive form of $C$ to an active form (or vice versa), Michaelis-Menten kinetics are assumed. Often, a given substance is scaled so that its total concentration is 1. In that case, there is a typical equation that describes the conversion from inactive protein (level $1 - C$) to the active form $C$:

$$\frac{\partial C}{\partial t} = \frac{K_1 E_{\text{activ}}(1 - C)}{J_1 + (1 - C)} - \frac{K_2 E_{\text{deactiv}} C}{J_2 + C} \tag{4.2}$$

Here $J_1, J_2$ are saturation constants and $K_1, K_2$ are the maximal rates of each of the two reactions and $E_i$ are the levels of the enzymes that catalyze the activation/inactivation. Such equations and expressions appear in numerous places in the models constructed by the Tyson group for the cell division cycle.

## 4.2 Hysteresis and biochemical switch in Cyclin and its antagonist

The simplest model is described by the diagram in Fig 4.1.




Figure 4.1: The simple model.

In this paper, Tyson and Novak first consider a system of two coupled ODE's for the mutual antagonism of Cyclin and proteolytic complex (APC), with equations

$$Y' = k_1 - (k_{2p} + k_{2pp}P)Y \tag{4.3}$$
$$P' = \frac{V_i P_i}{J_3 + P_i} - \frac{V_a P}{J_4 + P} \tag{4.4}$$

where $Y$=[CycB] = Cyclin-Cdk dimers,
$P$=[Cdh1]= active APC-Cdh1 complex,
$P_i$ = inactive APC complex.

The rates of the reactions are not constant. That is because a protein called here $A$ is assumed to enhance the forward reaction and cyclin ($Y$) enhances the reverse reaction. Tyson assumes that:

$$V_i = (k_{3p} + k_{3pp}A), \quad V_a = k_4 m Y$$

$$Y' = k_1 - (k_{2p} + k_{2pp}P)Y \tag{4.5}$$
$$P' = \frac{(k_{3p} + k_{3pp}A)P_i}{J_3 + P_i} - k_4 m \frac{YP}{J_4 + P} \tag{4.6}$$

It is assumed that the total amount of APC is constant, and scaled to 1: $P+P_i = 1$ so that

$$P_i = 1 - P$$

Hence,

$$Y' = k_1 - (k_{2p} + k_{2pp}P)Y \quad (4.7)$$

$$P' = \frac{(k_{3p} + k_{3pp}A)(1-P)}{J_3 + (1-P)} - k_4 m \frac{YP}{J_4 + P} \quad (4.8)$$

To make an AUTO bifurcation diagram, the system was first started with the parameter $m = 0.1$, and integrated for many time steps to arrive at the steady state $y = 0.038684p = 0.99402$. $m$ was used as the bifurcation parameter. Auto Axes were set as hI-lo, with Y on the Y-axis, and Main Parm:m on the horizontal axis, and with Xmin:0, Ymin:0, Xmax:0.6, Ymax:1.5.

This system was slightly fiddly, and a few first attempts at producing a bifurcation diagram with the default AUTO numerics parameters were unsuccessful (MX type error).

AutoNumerics parameters were adjusted as follows: *Ntst:50*, Nmax:200, NPr:50, *Ds:0.002, Dsmin: 0.0001*, Ncol:4, EPSL:0.0001, Dsmax:0.5, *Par Min: 0.1, Par Max: 0.2*, Norm min:0, Norm Max: 1000, EPSU: 0.001, EPSS:0,001.

The diagram was built up gradually by increasing the range of the plotted curve.

## 4.3 Including A=[Cdc20]

A new equation is introduced for an agent $A$ that represents [Cdc20]. That equation is given by:

$$A' = k_{5p} + k_{5pp}\frac{(mY/J_5)^n}{1 + (Ym/J_5)^n} - k_6 A$$

As a first analysis, $P$ is put on QSS from the previous model, leading to

$$P = G((k_{3p} + k_{3pp}A)/(k_4 m), Y, J_3, J_4)$$

where $G$ is a Goldbeter-Koshland function:

$$G(V_a, V_i, J_a, J_i) = \frac{2V_a J_i}{(V_i - V_a + V_a J_i + V_i J_a + \sqrt{(V_i - V_a + V_a J_i + V_i J_a)^2 - 4(V_i - V_a)V_a J_i}}$$

The equations of Model 2 are then

$$Y' = k_1 - (k_{2p} + k_{2pp}P)Y \quad (4.9)$$

$$A' = k_{5p} + k_{5pp}\frac{(mY/J_5)^n}{1 + (Ym/J_5)^n} - k_6 A \quad (4.10)$$

with $P$ given as above. The bifurcation diagram and a few phase plane portraits are shown in Figure 4.3. A subcritical Hopf bifurcation is found at $m = 0.5107$, giving birth to an unstable limit cycle. This is shown as an oval loop about the high SS value for $m = 0.5$ in the phase plane portrait in Fig 4.3.

## 4.4 The Y P A model

For the whole model, with no QSS assumption,

$$Y' = k_1 - (k_{2p} + k_{2pp}P)Y \tag{4.11}$$

$$P' = \frac{(k_{3p} + k_{3pp}A)(1-P)}{J_3 + (1-P)} - k_4 m \frac{YP}{J_4 + P} \tag{4.12}$$

$$A' = k_{5p} + k_{5pp}\frac{(mY/J_5)^n}{1 + (Ym/J_5)^n} - k_6 A \tag{4.13}$$

with the same decay and activation functions as before. Figure 4.4 shows the bifurcation diagram for this model. Note that this shares many features with the QSS version of Fig 4.3, but with somewhat different bifurcation values.

## 4.5 Addendum: A better AUTO XPP file

The following amended XPP file was produced by Dr. Anmar Khadra for a better AUTO bifurcation diagram.

```
#TysonAnmar.ode
# Equations
Y'=k1-(k2p+k2pp*P)*Y

P'=(1-P)*(k3p+k3pp*A)/(J3+1-P)-P*k4*m*Y/(J4+P)

A'=k5p+k5pp* ((m*Y/J5)^n)/(1+(y*m/J5)^n)-k6*A

# Parameters with units of 1/time:
p k1=0.04
p k2p=0.04,k2pp=1
p k3p=1,k3pp=10
p k4=35
p k5p=0.005,k5pp=0.2,k6=0.1

# mass of cell (try m=0.6, m=0.3, m=2)
p m=2

# Dimensionless parameters:
p J3=0.04,J4=0.04,J5=0.3,n=4
# Numerics
@ TOTAL=2000,DT=.1,xlo=0,xhi=2000,ylo=0,yhi=6
@ NPLOT=1,XP1=t,YP1=Y
@ MAXSTOR=10000000
@ BOUNDS=100000,METH=stiff
@ dsmin=1e-5,dsmax=.1,parmin=-.5,parmax=.5,autoxmin=-.5,autoxmax=.5
```

```
@ autoymax=.4,autoymin=-.5

# IC
Y(0)=1
P(0)=0.5
A(0)=0.1

done
```

Integrate the model (Initial conditions, Go, Initial cond, Last). Use the Data Viewer to find the period of the oscillation (in this case, 56.45). Got to nUmerics and change the Total time to this period, then escape and click I L I L I L. The curve should be superimposing on itself when you integrate for exactly one period.

The following settings were suggested for computing the full bifurcation diagram

Axes HiLo
Xmin 0.5
ymin 0
xmax 2
ymax 1
Numerics
Ntst 75
Nmax 2500
NPr 250
Ds -0.005
Dsmin 1e-07
Ncol 4
EPSL 1e-07
Dsmax 0.05
Par min 0.5
Par max 0.8
Norm min 0
Norm max 1e+07
EPSU 1e-07
EPSS 1e-07

The Figure 4.5 was produced with this ode file and settings.

## 4.6 Full Basic Model

Here we consider all variables in the model up to p 255 of Tyson's JTB paper, and the notation is:

$Y$=[CycB] = cyclin cdk dimers,
$P$=[Cdh1]= APC Cdh1 complex,

$A_T$= Total Cdc20 (not all of it in active form),
$A_A$ = active Cdc20,
$I_P$ = IAP, and $m$=cell mass.

$$
\begin{aligned}
Y' &= k_1 - (k_{2p} + k_{2pp}P)Y && (4.14)\\
P' &= \frac{(k_{3p} + k_{3pp}A_A)(1-P)}{J_3 + (1-P)} - k_4 m \frac{YP}{J_4 + P} && (4.15)\\
A_T' &= k_{5p} + k_{5pp}\frac{(mY/J_5)^n}{1 + (Ym/J_5)^n} - k_6 A_T && (4.16)\\
A_A' &= k_7 I_P \frac{A_T - A_A}{J_7 + A_T - A_A} - k_6 A_A - k_8[\text{Mad}]\frac{A_A}{J_8 + A_A} && (4.17)\\
I_P' &= k_9 mY(1 - I_P) - k_{10} I_P && (4.18)\\
m' &= \mu m \left(1 - \frac{m}{m_s}\right) && (4.19)
\end{aligned}
$$

Unlike the previous cases, we now assume that the activation function is

$$F_{activ}(P, A_A) = \frac{k_{3p} + k_{3pp}A_A}{J_3 + 1 - P}$$

that is, only the activated form of the Cd20 ($A_A$) will have an effect on activation of $P$. Also, the mass of the cell is halved every time that the cyclin concentration falls below a threshold value ($Y_{thresh} = 0.1$).

Simulations of this system with parameters given in the paper produce the figures shown in Fig 4.7.

## 4.7 Guide to the literature

Some of the elementary biochemical modules that were used to assemble larger regulatory networks are described in a popular article, (3). This has some of the description of the differential equations and their special properties. An entry-level review of cell-cycle modeling, (2), is a place to start if you want to avoid the details of the mathematical models, and get an informal biological introduction to the topic.

An article about the generic cell-cycle model is (4). The material described in this outline was drawn from that source, and from a readable account in a chapter written by Tyson in the book edited by Fall et al. (5). For students who have had some exposure to bifurcation analysis, and who want to find out more about the interesting behaviour of the cell cycle model and its bifurcations, a good reference is (6). Here you will find a compendium of the types of bifurcations encountered (with an explanantory Appendix).

The Tyson Lab website (1) contains a link to downloadable files, including XPP .ode files that can be used to run and simulate a variety of published models.




Figure 4.2: Top: The YP phase plane for parameters as shown in the XPP file. Parameters with units of 1/time are: $k_1 = 0.04, k_{2p} = 0.04, k_{2pp} = 1, k_{3p} = 1, k_{3pp} = 10, k_4 = 35$, Other (dimensionless) parameters are: $A = 0, J_3 = 0.04, J_4 = 0.04$ Bottom: a bifurcation diagram produced with XPP Auto. The parameter $m$, the cell mass, is the bifurcation parameter.




Figure 4.3: Bifurcation diagram for model 2. An unstable limit cycle occurs over the range $0.4962 < m < 0.5107$. Phase plane portraits are shown for $m = 0.3, 0.5, 0.6$. Some of the trajectories on these were computed by setting $\Delta t$ to be a small negative timestep, i.e. integrating backwards in time.




Figure 4.4: Bifurcation diagram for model 2 with no QSS. There is a sub-critical HOpf bifurcation at $m = 0.5719$. An unstable limit cycle occurs over the range $0.4706 < m < 0.5719$. There is a range of parameters below this value for which three steady states exist, two of which are stable.




Figure 4.5: The stable limit cycle for larger values of $m$. This was obtained with the .ode file TysonAnmar.ode




Figure 4.6: The full model.




Figure 4.7: Time behaviour of variables in full model.

# Bibliography

http://mpf.biol.vt.edu/Research.html

Tyson JJ, Csikasz-Nagy A, Novak B. (2002) The dynamics of cell cycle regulation. Bioessays 24(12):1095-109.

Tyson JJ, Chen KC, Novak B. (2003) Sniffers, buzzers, toggles and blinkers: dynamics of regulatory and signaling pathways in the cell. Curr Opin Cell Biol. 15(2):221-31.

Tyson JJ, Novak B. (2001) Regulation of the eukaryotic cell cycle: molecular antagonism, hysteresis, and irreversible transitions. J Theor Biol. 210(2):249-63.

Fall CP, Marland ES, Wagner JM, Tyson JJ (2002) **Computational Cell Biology**, Springer, NY.

Csikasz-Nagy A, Battogtokh D, Chen KC, Novak B, Tyson JJ. (2006) Analysis of a generic model of eukaryotic cell-cycle regulation. Biophys J. 90(12):4361-79.

## 4.8 Appendix: XPP codes

### 4.8.1 Code for simplest 2-variable model

The following code produced Fig 4.2.

```
# tysonCCJTB01.ode

# Model based on first system (eqs 2) shown in Tyson's paper
# JTB (2001) vol 210 pp 249-263

# Y=[CycB] = cyclin cdk dimers
# P=[Cdh1]= APC Cdh1 complex (proteolytic complex)
# Y and P are mutually antagonistic

Y'=k1-(k2p+k2pp*P)*Y
P'=Factiv(P)*(1-P)-Fdecay(Y,P)*P

Factiv(P)=(k3p+k3pp*A)/(J3+1-P)
Fdecay(Y,P)=k4*m*Y/(J4+P)

# parameters with units of 1/time:
par k1=0.04
par k2p=0.04,k2pp=1
par k3p=1,k3pp=10
par k4=35

par A=0

# mass of cell (try m=0.6, m=0.3)
par m=0.3

# Dimensionless parameters:
par J3=0.04,J4=0.04

@ dt=0.005
@ xp=Y,yp=P,xlo=0,xhi=1,ylo=0,yhi=1
done
```

### 4.8.2 Code for model with QSS on P

The following code produced Fig 4.3.

```
# tysonCCJTB01_3QSSP.ode

# Model based on three eqns system (eqs 3) shown in Tyson's paper
# JTB (2001) vol 210 pp 249-263
```

```
# Y=[CycB] = cyclin cdk dimers
# P=[Cdh1]= APN Cdh1 complex (proteolytic complex)
# Y and P are mutually antagonistic
# A= Cdc14=Cdc20 - eqn (3) in this paper
# In this file we put P on QSS and look at Y and A

Y'=k1-(k2p+k2pp*P)*Y

#P'=Factiv(P)*(1-P)-Fdecay(Y,P)*P
#Factiv(P)=(k3p+k3pp*A)/(J3+1-P)
#Fdecay(Y,P)=k4*m*Y/(J4+P)

P=G((k3p+k3pp*A)/(k4*m),Y,J3,J4)
G(Va,Vi,Ja,Ji)=2*Va*Ji/(Vi-Va+Va*Ji+Vi*Ja+ sqrt((Vi-Va+Va*Ji+Vi*Ja)^2-4*(Vi-Va)*Va*Ji))

A'=k5p+k5pp* ((m*Y/J5)^n)/(1+(y*m/J5)^n)-k6*A

# parameters with units of 1/time:
par k1=0.04
par k2p=0.04,k2pp=1
par k3p=1,k3pp=10
par k4=35
par k5p=0.005,k5pp=0.2,k6=0.1


# mass of cell (try m=0.6, m=0.3)
par m=0.5

# Dimensionless parameters:
par J3=0.04,J4=0.04,J5=0.3,n=4

@ dt=0.005
@ xp=A,yp=Y,xlo=0,xhi=1,ylo=0,yhi=1
done
```

### 4.8.3 Code for 3-variable model

The following code produces Figure 4.4 in Section 4.4.

```
# tysonCCJTB01_3.ode

# Model based on three eqns system (eqs 3) shown in Tyson's paper
# JTB (2001) vol 210 pp 249-263

# Y=[CycB] = cyclin cdk dimers
```

```
# P=[Cdh1]= APN Cdh1 complex (proteolytic complex)
# Y and P are mutually antagonistic
# A= Cdc14=Cdc20 - eqn (3) in this paper

Y'=k1-(k2p+k2pp*P)*Y
P'=Factiv(P)*(1-P)-Fdecay(Y,P)*P

Factiv(P)=(k3p+k3pp*A)/(J3+1-P)
Fdecay(Y,P)=k4*m*Y/(J4+P)


A'=k5p+k5pp* ((m*Y/J5)^n)/(1+(y*m/J5)^n)-k6*A

# parameters with units of 1/time:
par k1=0.04
par k2p=0.04,k2pp=1
par k3p=1,k3pp=10
par k4=35
par k5p=0.005,k5pp=0.2,k6=0.1


# mass of cell (try m=0.6, m=0.3)
par m=0.3

# Dimensionless parameters:
par J3=0.04,J4=0.04,J5=0.3,n=4

@ dt=0.005
done
```

### 4.8.4 Code for the full model (Fig 4.7)

```
# tysonCCJTB01_4.ode

# Model based on Full model (eqs 1-6) shown in Tyson's paper
# JTB (2001) vol 210 pp 249-263

# Y=[CycB] = cyclin cdk dimers
# P=[Cdh1]= APN Cdh1 complex (proteolytic complex)
# Y and P are mutually antagonistic
# A= Cdc14=Total Cdc20 - eqn (3) in this paper
#AA = active Cdc20 = Cdc20_A eqn (4)
# IP = [IAP]  eqn(5)

Y'=k1-(k2p+k2pp*P)*Y
P'=Factiv(P)*(1-P)-Fdecay(Y,P)*P
```

```
######NOTE: CHANGE IN THE FOLLOWING FORMULA A->AA
# as per Tyson's discussion on p 255

Factiv(P)=(k3p+k3pp*AA)/(J3+1-P)
Fdecay(Y,P)=k4*m*Y/(J4+P)


A'=k5p+k5pp* ((m*Y/J5)^n)/(1+(y*m/J5)^n)-k6*A

AA' = k7*IP*(A-AA)/(J7+A-AA) -k6*AA -k8*Mad*AA/(J8+AA)

IP'=k9*m*Y*(1-IP)-k10*IP

m'= mu*m*(1-m/ms)

# parameters with units of 1/time:
par k1=0.04
par k2p=0.04,k2pp=1
par k3p=1,k3pp=10
par k4=35
par k5p=0.005,k5pp=0.2,k6=0.1

par k7=1,k8=0.5,Mad=1
par k9=0.1,k10=0.02
par mu=0.01,ms=10

#Global flag: See XPP book p 36
# When Y falls below threshold, the cell divides,
# then its mass m is 1/2 its previous mass.
global -1 Y-Ythresh {m=m/2}
par Ythresh=0.1

# mass of cell (try m=0.6, m=0.3)
#par m=0.3

# Dimensionless parameters:
par J3=0.04,J4=0.04,J5=0.3,n=4
par J7=0.001,J8=0.001


init Y=0.6,P=0.02,A=1.6,AA=0.6,IP=0.6,m=0.8

@ dt=0.005,Total=300,MAXSTOR=500000
done
```